
	下面是小编整理的一份Excel公式，平时会计用的最多。如果你身边有做会计的朋友，你可以转发分享给他们哦，也许会对他们有所帮助!



	总共有20个Excel公式，其中还包括2019最新个税公式。下面就随PDF帝小编一起来看下吧!



	1、文本与百分比连接公式 



	如果直接连接，百分比会以数字显示，需要用Text函数格式化后再连接



	="本月利润完成率为"&TEXT(C2/B2,"0%")

	2、账龄分析公式 



	用lookup函数可以划分账龄区间



	=LOOKUP(D2,G$2:H$6)



	如果不用辅助区域，可以用常量数组



	=LOOKUP(D2,{0,"小于30天";31,"1~3个月";91,"3~6个月";181,"6-1年";361,"大于1年"})

	3、屏蔽错误值公式 



	把公式产生的错误值显示为空



	公式：C2



	=IFERROR(A2/B2,"")



	说明：如果是错误值则显示为空，否则正常显示。

	4、完成率公式 



	如下图所示，要求根据B的实际和C列的预算数，计算完成率。



	公示：E2



	=IF(C3<0,2-B3/C3,B3/C3)

	5、同比增长率公式 



	如下图所示，B列是本年累计，C列是去年同期累计，要求计算同比增长率。



	公示：E2



	=(B2-C2)/IF(C2>0,C2,-C2)

	6、金额大小写公式 



	=TEXT(LEFT(RMB(A2),LEN(RMB(A2))-3),"[>0][dbnum2]G/通用格式元;[<0]负[dbnum2]G/通用格式元;;")&TEXT(RIGHT(RMB(A2),2),"[dbnum2]0角0分;;整")

	7、单条件查找公式 



	公式1：C11



	=VLOOKUP(B11,B3:F7,4,FALSE)

	8、双向查找公式 



	公式：



	=INDEX(C3:H7,MATCH(B10,B3:B7,0),MATCH(C10,C2:H2,0))



	说明：利用MATCH函数查找位置，用INDEX函数取值

	9、多条件查找公式 



	公式：C35



	=Lookup(1,0/((B25:B30=C33)*(C25:C30=C34)),D25:D30)

	10、单条件求和公式 



	公式：F2



	=SUMIF(A:A,E2,C:C)

	11、多条件求和公式 



	=Sumifs(c2:c7,a2:a7,a11,b2:b7,b11)

	12、隔列求和公式 



	公式H3:



	=SUMIF($A$2:$G$2,H$2,A3:G3)

	如果没有标题，那只能用稍复杂的公式了。



	=SUMPRODUCT((MOD(COLUMN(B3:G3),2)=0)*B3:G3)



	13、多条件判断公式 



	公式：C2



	=IF(AND(A2<500,B2="未到期"),"补款","")



	说明：两个条件同时成立用AND,任一个成立用OR函数。

	14、多表相同位置求和 



	公式：b2



	=SUM(Sheet1:Sheet19!B2)



	说明：在表中间删除或添加表后，

	公式结果会自动更新。



	15、两条查找相同公式 



	公式:B2



	=COUNTIF(Sheet15!A:A,A2)



	说明：如果返回值大于0说明在另一个表中存在，0则不存在。

	16、两表数据多条件核对 



	如下图所示，要求核对两表中同一产品同一型号的数量差异，显示在D列。



	公式：D10



	=SUMPRODUCT(($A$2:$A$6=A10)*($B$2:$B$6=B10)*$C$2:$C$6)-C10

	17、个税计算 



	工资表调整



	工资表原个税列变为3列，分别是累计应缴预扣所得额、累计税额、本月应扣缴税额。(列标题大家自已命名吧)



	




	[image: ] 


	公式设置



	1月工资表公式：



	D2(累计所得额)



	=B2-C2-5000



	E2(本月累计税额)



	=5*MAX(0,D2*{0.6;2;4;5;6;7;9}%-{0;504;3384;6384;10584;17184;36384})



	F2(本月应扣缴额)



	=E2

	2月工资表公式：



	D2(累计所得额)



	=B2-5000-C2+IFERROR(VLOOKUP(A2,'1月工资表'!A:D,4,0),0)



	注：使用vlookup查找上月累计应扣应缴所得额



	E2(本月累计税额)



	=5*MAX(0,D2*{0.6;2;4;5;6;7;9}%-{0;504;3384;6384;10584;17184;36384})



	注：公式同上月



	F2(本月应扣缴额)



	=E2-IFERROR(VLOOKUP(A2,'1月工资表'!A:F,5,0),0)



	注：本月应扣 = 本月累计应扣应缴个税 - 上月累计数

	其他月份工资表同2月，只需要把公式中引用的工作表改为上一月即可。如制作4月份工资表，公式中的2月改为3月即可。



	18、多表同一位置汇总公式 



	多个工作表如果格式完全相同，可以用sum函数的多表求和功能。



	如下图所示，要求在汇总表里设置合计公式，汇总前19个工作表B列的和。



	B2 =SUM(Sheet1:Sheet19!B2)

	19、中国式排名公式 



	公式:C12



	=SUMPRODUCT(($D$4:$D$9>=D4)*(1/COUNTIF(D$4:D$9,D$4:D$9)))

	20、先进先出公式 



	提醒：该公式属超高难度公式，不建议新手使用和费力的去理解，仅供excel中高水平用户参考使用。



	定义名称：



	Lj



	=SUBTOTAL(9,OFFSET(!$E$2,,,ROW()-1))-SUBTOTAL(9,OFFSET(!$B$2,,,ROW(!$B$1:$B9)))



	G3公式：



	{=SUMPRODUCT(IF(lj>$B$3:B3,$B$3:B3,lj),IF(lj>0,$C$3:C3))-SUM($G$2:G2)}

	如果已经掌握了上面的函数公式，在数据处理中常见的excel问题基本上都可以解决了，当然，如果是比较复杂的问题，还是需要掌握另外20%的高难度公式。

